
Appendix G. ANS to TOF extension


TOF contains many non-ANS words that are used for more efficient execution on small processors.  An ANS compliant Forth can be used to test TOF code by loading the file ANS_TOF.F.  It was tested using Win32forth, which itself contains some non-ANS words that may not be covered here.

When you’re trying to write portable Forth code, the question isn’t “Is this ANS”?  The question is, “What will it take for an ANS Forth to handle this”?

ANS_TOF.F is listed here, with commentary inserted.

MULTI..WHILE..REPEAT

\ MULTI..WHILE..REPEAT provides a means to do something zero or more times.

\ For example, 4 >R MULTI R@ REPEAT R>DROP puts 3 2 1 0 on the stack.

\ It translates cleanly to machine code: bump the cell on top of the return

\ stack and branch conditionally.

\ Literal >R and R>DROP also reduce to small code sequences.

: MULTI ( -- )        ( -- R: x -- x' )

        postpone BEGIN

        postpone R>  postpone 1-  postpone DUP  postpone >R

        postpone 0<  postpone 0=

        postpone WHILE ; immediate

Since ?DO does bounds checking (among other things) ?DO ... LOOP is a very inefficient way to do something zero or more times.  MULTI ... REPEAT bumps the top of the return stack downward to zero, so R@ can be used as an index or count value if needed.  

Native code to perform this loop structure can be very small.  The 68000 TOF kernel uses the following definition for _TYPE, which represents MULTI with two instructions.

: _TYPE ( addr len --)

     >r multi   count emit

        repeat  r>drop drop ;

00021802 2F07         _TYPE:   MOVE.L D7,-(A7)


\ >R

00021804 2E1D                  MOVE.L (A5)+,D7

00021806 5397                  SUBQ.L #1,(A7)


\ MULTI

00021808 6B000012              BMI 0x2181C

0002180C 4EB90004FD60          JSR 0x4FD60(xt:112) COUNT

00021812 4EB90004FB80          JSR 0x4FB80(xt:192) EMIT

00021818 6000FFEC              BRA 0x21806



\ REPEAT

0002181C 588F                  ADDQ.L #4,A7



\ R>DROP

0002181E 2E1D                  MOVE.L (A5)+,D7


\ DROP

00021820 4E75                  RTS

The AVR builder has an optimization that reduces a literal followed by >R to four machine instructions.  Similar optimizations can be done on most processors.

MACHINE FORTH POINTERS

\ The A register is an idea adapted from machine Forth. Having an A register

\ is very handy for indexing and temporary storage.

variable A

: A!    ( addr -- )     A ! ;                      \ set A register

: A@    ( -- addr )     A @ ;                      \ get A register

: @A    ( -- x ) A @ @ ;                           \ fetch cell from A stream

: @A+   ( -- x ) A @ @  [ 1 CELLS ] LITERAL A +! ; \ lift cell from A stream

: C@A   ( -- c ) A @ C@ ;                          \ fetch char from A stream

: C@A+  ( -- c ) A @ C@ [ 1 CHARS ] LITERAL A +! ; \ lift char from A stream

: !A    ( x -- ) A @ ! ;                           \ store cell to A stream

: !A+   ( x -- ) A @ !  [ 1 CELLS ] LITERAL A +! ; \ append cell to A stream

: C!A   ( x -- ) A @ C! ;                          \ store char to A stream

: C!A+  ( x -- ) A @ !  [ 1 CHARS ] LITERAL A +! ; \ append char to A stream

: @R    ( -- x ) ( R: a -- a )    R@ @ ;           \ fetch cell from R stream

: @R+   ( -- x ) ( R: a -- a+4 )  R@ @             \ lift cell from R stream

                     R> [ 1 CELLS ] LITERAL + >R ;

: !R    ( x -- ) ( R: a -- a )    R@ ! ;           \ store cell to R stream

: !R+   ( x -- ) ( R: a -- a+4 )  R@ !             \ append cell to R stream

                     R> [ 1 CELLS ] LITERAL + >R ;

BIT TESTING

\ IFSET and IFCLR perform non-destructive bit tests on the top of the stack.

\ They are good for parsing a packed field of bit flags. On processors with bit

\ tests, this produces efficient in-line code.

\ IFSET and IFCLR take a parameter at compile time:

\ [ 3 ] IFSET MyStuff THEN is the same as DUP 8 AND IF MyStuff THEN

: IFSET ( bit# -- )   ( n -- n )

        1 swap lshift

        postpone LITERAL

        postpone OVER postpone AND

        postpone IF ; immediate

: IFCLR ( bit# -- )   ( n -- n )

        1 swap lshift invert

        postpone LITERAL

        postpone OVER postpone AND

        postpone IF ; immediate

These are good for decoding bit flags in a packed field.  A few machine instructions are all that’s necessary to encode one of these structures.  Some targets are limited to 8-bit jumps, so be careful not to put a lot of code between IFCLR or IFSET and THEN.

COMPACT CASE STATEMENT

\ The [ n ] ?[ ... ]? structure conditionally executes a snippet of code.

\ It compares A to a literal and executes if it's a match. Then it exits.

: QCASE: ( -- )  ( c -- )

        postpone A! ; immediate

: ?[    ( c -- )

        postpone LITERAL

        postpone A@ postpone =

        postpone IF ; immediate

: ]?    ( -- )

        postpone EXIT

        postpone THEN ; immediate

QCASE: loads an 8-bit value into a register for subsequent tests by ?[.  Since some Forth words could trash this register, there are some restrictions on usage.  All ?[ ... ]? groups should be grouped together, with the first immediately following QCASE:. 

This can best be illustrated by the following example:

: test          ( c -- )

         qcase: [ 1 ] ?[ sqrt ]?

                [ 3 ] ?[ dist ]?

                [ 5 ] ?[ dup over rot ]?

                ;

00A64 2F0A     TEST:    MOV   R16,R26

\ Load R16 with test character (AVR)

00A66 91B9              LD    R27,Y+

00A68 91A9              LD    R26,Y+

00A6A 3001              CPI   R16,1

\ [ 1 ] ?[ 

00A6C F409              BRNE  0xA70

00A6E CFB5              RJMP  0x9DA:SQRT
\ sqrt  ]?

00A70 3003              CPI   R16,3

\ [ 3 ] ?[ 

00A72 F409              BRNE  0xA76

00A74 CFDE              RJMP  0xA32:DIST
\ dist ]?

00A76 3005              CPI   R16,5

\ [ 5 ] ?[ 

00A78 F421              BRNE  0xA82

00A7A 93AA              ST    -Y,R26

\ dup 

00A7C 93BA              ST    -Y,R27

00A7E DADA              RCALL 0x34:OVER

\ over

00A80 CD64              RJMP  0x54A:ROT   
\ rot ]?

00A82 9508              RET

The branches and comparisons built by the qcase structure aren’t broken up by Forth calls.  As with IFSET, branches can be 8-bit so there will be size restrictions on code within ?[ .. ]?.

The QCASE structure lays down multiple exit points, so it’s usually the last thing in a definition.

MEMORY OPERATORS

\ Other TOF kernel words not in ANS or Win32forth:

: C@P           ( addr -- c )   C@ ;    \ char fetch from program memory

: @P            ( addr -- x )   @ ;     \ fetch from program memory

: W@P           ( addr -- x )   W@ ;    \ 16bit fetch from program memory

: C!P           ( c addr -- )   C! ;    \ char store to program memory

: !P            ( x addr -- )   ! ;     \ store to program memory

: W!P           ( x addr -- )   W! ;    \ 16bit store to program memory

: PCREATE       ( <name> -- )   create ;

In Forth, code and data memory spaces aren’t required to overlap. We assume the ANS Forth that we’re extending overlaps code space and data space. Forth operators typically operate on data space. TOF sometimes copies data from ROM to RAM at startup to accommodate this. TOF has special operators to access program space, since data memory and program memory aren’t necessarily the same thing.

If you want to create a table in ROM, you should use PCREATE <name> to create it and @P etc. to read it.  These operators work on code (program) space, not data space.  A data structure built with CREATE uses startup code to initialize RAM from ROM.  The data structure is then free to be modified.  If this flexibility isn’t needed, using PCREATE will save ROM and RAM.

BOOLEAN TESTS

\ Variables used as Boolean flags are more readable with Boolean-like tests:

: ON?           ( addr -- f )   @ 0<> ; \ flag at addr <> 0?

: OFF?          ( addr -- f )   @ 0= ;  \ flag at addr = 0?

These are the complements of ON and OFF, which operate on cell-wide Booleans. For more natural readability, I like to use variable names that are adjectives or end in ing.

\ Embedded micros can efficiently test bits in memory. So...

: (setupbit)    ( bit# addr -- addr n mask )  DUP >R  1 SWAP LSHIFT  SWAP R> C@ ;

: BIT-ON        ( bit# addr -- )        \ set bit# of char at addr

                (setupbit) OR SWAP C! ;

: BIT-OFF       ( bit# addr -- )        \ clear bit# of char at addr

                (setupbit) INVERT AND SWAP C! ;

: BIT?          ( bit# addr -- f )      \ test bit# of char at addr

                (setupbit) AND NIP 0<> ;

These operators work on bit variables in memory.  They take advantage of the bit tests provided by most microcontrollers.  Bit numbers 0..7 are valid, with 7 being the MSB of a byte.  

Note that if you have some control over how the hardware is designed, you can assign each I/O pin its own cell address (use one bit per cell) that’s compatible with ON and OFF.

MIXED AND MISC. ARITHMETIC 

: M/MOD         ( d n -- r q )          ( floored )     \ signed   d/n --> r q

                DUP 0<  DUP>R

                IF      NEGATE >R DNEGATE R>

                THEN    >R DUP 0<

                        IF      R@ +

                        THEN    R> UM/MOD R>

                IF      SWAP NEGATE SWAP

                THEN    ;

: MU/MOD        ( ud# un1 -- rem d#quot )               \ unsigned ud/u --> ur udq

                >R  0  R@  UM/MOD  R>  SWAP

                >R  UM/MOD  R> ;

: U/MOD         ( u1 u2 -- r q )  DROP UM/MOD ;         \ unsigned u/u --> r q

: UD2/          ( d -- d/2 )    

                d2/ [ -1 1 rshift ] literal and ;       \ strip MSB

: >>A           ( x count -- )                          \ arithmetic right shift

                -1 OVER RSHIFT INVERT >R  OVER 0< >R

                RSHIFT R> R> AND OR ;

: D0<>          ( d -- f )  D0= 0= ;                    \ true if double-cell <> 0

OTHER OPERATORS

\ Openboot has COMP, which is the F83 compare, not the ANS compare

: COMP          ( a1 a2 len -- f )   TUCK COMPARE ;

: UNDER1+       ( x1 x2 -- x1' x2 )  >R 1+ R> ;         \ add 1 to NOS

: UNDER1-       ( x1 x2 -- x1' x2 )  >R 1- R> ;         \ subtract 1 from NOS

: >DIGIT        ( n -- c ) DUP 9 > 7 AND + [CHAR] 0 + ; \ convert digit to ASCII

hex

: C>N           ( c -- n )      dup   80 and 0<>   -80 and or ;  \ sign extend

: W>N           ( c -- n )      dup 8000 and 0<> -8000 and or ;

: BYTE-SPLIT    ( n -- cl ch )                          \ split into lo and hi bytes

                DUP 0FF AND SWAP 8 RSHIFT 0FF AND ;

: BYTE-JOIN     ( cl ch -- n )  8 LSHIFT OR ;           \ join lo and hi bytes

: BYTE-SWAP     ( n -- n' ) BYTE-SPLIT SWAP BYTE-JOIN ; \ swap lower 2 bytes of n

: WORD-JOIN     ( nl nh -- n ) 10 LSHIFT OR ;           \ join  lo-16 hi-16 --> 32

: WORD-SPLIT    ( n -- nl nh )                          \ split 32 --> lo-16 hi-16

                DUP 0FFFF AND SWAP 10 RSHIFT 0FFFF AND ;

decimal

: LW,           ( n -- ) byte-split swap c, c, ;        \ comma 16-bit little endian

: C(            postpone ( ; immediate                  \ catalog (summary) comment

: CVARIABLE     variable ;                              \ allots space for 1 character

: {{ ; : }} ; \ ignore {{ }}

variable debugging

: \D debugging @ 0= if postpone \ then ; immediate

: \ND debugging @   if postpone \ then ; immediate

These are in the HOME wordlist. To enable debugging, you can use {{ debugging on }} anytime.  Use \D to (sometimes) comment out debugging code.
: timestamp     ( -- n )

\ Get 32-bit time stamp starting at Jan 1st, 2000.  1-second resolution.

                time&date 2000 -        ( s m h d m y )

                12 * swap +             \ Rolls over every 133 years.

                31 * swap +

                24 * swap +

                60 * swap +

                60 * + ;

You can use this to timestamp the ROM image.

FRACTIONAL MATH

: uf1.0         ( -- d )
0 1 ;
\ unsigned 1.0

: sf            ( <number> -- n )

\ get signed fractional number ( -1.0 to +1.0 ) from input stream,

\ convert to signed integer ( minint .. maxint )

                bl parse >float 0= abort" Expecting fractional number .XXXXX"

                fdup -1e0 f<

                fdup  1e0 f> or abort" Range must be -1.0 .. +1.0"

                uf1.0 d>f 2e0 f/ f* f>d 0=

                if      dup uf1.0 d2/ drop = if 1- then \ clip +1.0 to maxint

                then    ;

: uf            ( <number> -- n )

\ get unsigned fractional number ( 0 to +1.0 ) from input stream,

\ convert to umsigned integer ( 0 .. umaxint )

                bl parse >float 0= abort" Expecting fractional number .XXXXX"

                fdup  0e0 f<

                fdup  1e0 f> or abort" Range must be -1.0 .. +1.0"

                uf1.0 d>f f* f>d

                if      1-              \ clip +1.0 to umaxint

                then    ;

These operators are used to represent fractional numbers in a cell-width independent way.  A value between 0 and 1 scales to an integer between 0 and 2cellwidth if unsigned.  Signed arithmetic represents a value between –1 and 1 as an integer between -2cellwidth-1 and 2cellwidth-1.  

In each case, the integer version of +1.0 is bumped down by 1 to avoid an overflow condition.  For example, signed +1.0 is changed to +0.99997 with 16-bit cells, or +0.9999999995 with 32-bit cells.  If this presents an accuracy problem, scale so as to avoid +1.0.

